
 12.1   資料整理的常用方法 

Excel 工作表中可以存放眾多資料，利用 Excel 提供的方法可以快速搜

尋、排序、篩選出想要的資料，因此大大提高工作的效率。 

12.1.1  SpecialCells 方法 

使用 Range 物件的 SpecialCells 方法，會傳回儲存格範圍內和指定的類

型及值相符的所有儲存格，傳回值 r 為 Range 物件，其語法為： 

   語法   

r = 儲存格範圍.SpecialCells(Type[, Value]) 

說明  

1. Type 引數為必要的引數是指定傳回儲存格類型，常用的引數值有

xlCellTypeConstants(含常值)、xlCellTypeFormulas(含公式)、

xlCellTypeVisible(可見)、xlCellTypeBlanks(空白)、xlCellTypeLastCell 

(右下角儲存格)、xlCellTypeAllFormatConditions(設定格式)...。 

2. 如果 Type 引數值為 xlCellTypeConstants 或 xlCellTypeFormulas 時，可用

Value 引數進一步指定儲存格類別。Value 引數值有：xlNumbers(數值)、

xlTextValues(字串)、xlLogical(邏輯值)、xlErrors(錯誤值)。 

【例 1】將使用儲存格範圍內的空白儲存格設定值為 0： 

UsedRange.SpecialCells(xlCellTypeBlanks).Value = 0 

【例 2】將使用儲存格範圍內含公式且值為錯誤的儲存格，設為紅色字： 

UsedRange.SpecialCells(xlCellTypeFormulas, xlErrors).Font.Color = vbRed 

【例 3】加總儲存格範圍內含數值的儲存格： 

total = Application.SUM(UsedRange.SpecialCells(xlCellTypeConstants, _ 
                                       xlNumbers)) 

12.1.2  Sort 方法 

Excel 的工作表可以做為一個簡易的資料庫，只要將工作表的第一列建

立為資料的欄位 (標題)，第二列起為各個欄位的資料值即可。使用 Range 物

件的 Sort 方法會針對指定儲存格範圍，根據指定的條件進行資料的排序，其

語法為： 

   語法   

儲存格範圍.Sort([ Key1] [, Order1] [, Key2] [, Type] [, Order2] _ 
[, Key3] [, Order3] [, Header] [, OrderCustom] [, MatchCase] _ 
[, Orientation] [,SortMethod] [, DataOption1] [, DataOption2] _ 
[, DataOption3]) 

說明  

1. Key1 引數是指定第一個排序依據欄位(主要鍵)，通常 Key 引數都是指定標

題的儲存格或欄位名稱。Key2 引數是第二個排序依據欄位(次要鍵)，Key3
引數是第三個排序依據欄位(第三鍵)。 

2. Order1 引數是指定主要鍵的排序順序，引數值有 xlAscending (遞增，預設

值)、xlDescending (遞減)。Key1 搭配 Order1 成為一組引數，Key2、Key2
則分別搭配 Order2 和 Order3。 

3. Header引數是設定是否有標題列，引數值有 xlNo(沒有標題，預設值)、xlYes 

(有標題)、xlGuess (由系統判斷)。 

4. MatchCase 引數設定是否區分英文字母大小寫，引數值有 True(區分)、

False(不區分)。 

5. Orientation 引數是設定排序方向，引數值有 xlSortRows(列，預設值)、

xlSortColumns(欄)。 

6. SortMethod 引數是設定中文字的排序依據，引數值有 xlPinYin(注音，預

設值)、xlStroke(筆畫)。 

7. DataOption1 引數是設定 Key1 引數的文字和數字排序規則，引數值有

xlSortNormal (數字和文字資料分開，預設值) 、xlSortTextAsNumbers (文
字視為數字)。DataOption2、DataOption3 引數分別搭配 Key2、Key3 引數。 

8. 使用 Sort 方法後，系統會保留 Header、Order1、Order2、Order3、

OrderCustom 和 Orientation 等引數的設定值，供下次使用 Sort 方法時延

用。如果要避免前次保留值影響排序結果時，就要完整指定這些引數值。 

9. Rang 物件的 Sort 方法最多只能指定 3 個排序欄位，當需要排序的欄位超

過 3 個時，就可以執行 Sort 方法多次來達成。 

10.執行 Sort 方法時工作表畫面會隨之不斷異動，想加快速度可以在執行前，

先將 Application 物件的 ScreenUpdating 屬性值設為 False，執行 Sort 方法

後再重設為 True。 

【例 1】A1:D10 儲存格範圍根據 A 欄做遞減排序(有標題列)： 

Range("A1:D10").Sort Key1:=Range("A1"), Order1:=xlDescending, _ 
                                                 Header:=xlYes 

【例 2】A 到 E 欄主要依 B 欄遞增，次要依 D 欄遞減排序(無標題列)： 

Columns("A:E").Sort Key1:=Columns("B"), Order1:=xlAscending, _ 
                 Key2:=Columns("D"), Order2:=xlDescending, Header:=xlNo 

【例 3】A 到 G 欄依 B、C、D、E 欄的優先順序做遞增排序： 

Columns("A:G").Sort Key1:=Range("D1"), Key2:=Range("E1") 
'先處理優先順序較低的欄位，優先順序高的後處理 
Columns("A:G").Sort Key1:=Range("B1"), Key2:=Range("C1") 

【例 4】在標題欄上快按兩下，就依照該欄做遞增排序： 

Private Sub Worksheet_BeforeDoubleClick(ByVal Target As Range, _ 
                                                                   Cancel As Boolean) 
    Dim dataRng As Range 
    Set dataRng = UsedRange                  ' 取得資料儲存格範圍 
    Dim colCount As Integer 
    colCount = dataRng.Columns.Count   ' 取得資料儲存格範圍的欄位數 

'若是第一列欄位 
    If Target.Row = 1 And Target.Column <= colCount Then  
        Cancel = True 
        dataRng.Sort Key1:=Target, Header:=xlYes      '依 Target 欄位排序 
    End If 
    Set dataRng = nothing 
End Sub 

12.1.3  AutoFilter 方法 

使用 Range 物件的 AutoFilter 方法，可以根據準則篩選工作表中的資料，

將不符合條件的資料以整列方式隱藏，只顯示出符合條件的資料。就像是執

行「自動篩選」功能，其語法為： 

   語法   

儲存格範圍.AutoFilter([Field][, Criteria1][, Operator][, Criteria2] _ 
                        [, VisibleDropDown]) 

說明  

1. Field 引數是指定篩選依據的欄位，引數值為整數。如果要指定 A 欄引數

值應設為 1，B 欄為 2…依此類推。 

2. Criteria1 引數是指定篩選的準則，引數可為數值、字串、陣列等，可搭配

Operator 引數。例如"通過"、"電*"(以電字開頭)、"<0"(數值小於 0) 、 

"="(空白儲存格)、"<>"(非空白儲存格)、"><"(無資料儲存格)，省略時預

設為全部。 

3. Operator 引數搭配 Criteria1 引數指定篩選的類型，或兩個篩選準則的邏輯

運算。常用的引數值為 xlAnd(Criteria1 和 Criteria2 篩選準則做 AND 邏輯

運算)、xlOr(Criteria1 和 Criteria2 篩選準則做 OR 邏輯運算)、xlTop10Items 

(顯示 Criteria1 指定數量的最高值項目)、xlBottom10Items(顯示 Criteria1
指定數量的最低值項目)、xlTop10Percent(顯示 Criteria1 指定百分比的最

高值項目)、xlBottom10Percent(顯示 Criteria1 指定百分比的最低值項目)、

xlFilterValues(條件值)、xlFilterCellColor(儲存格的色彩)、xlFilterFontColor 

( 字型的色彩 ) 、 xlFilterDynamic( 動態篩選 ) 。例如 Operator 引數值為

xlTop10Items，Criteria1 引數值為 3，表篩選出前三大的資料。 

4. Criteria2 引數是指定第二個篩選準則，搭配 Criteria1 及 Operator 引數。 

5. VisibleDropDown 引數設定是否顯示篩選欄位的下拉鈕，引數值有 True(顯
示，預設值)、False(隱藏)。 

6. 若 AutoFilter 方法沒有任何引數值時，只會切換篩選下拉鈕的顯示狀態，

可用來取消自動篩選。工作表使用自動篩選時，該工作表的 AutoFilterMode
屬性值為 True，若設為 False 可取消自動篩選。使用工作表的 ShowAllData
方法會顯示全部資料，但仍屬自動篩選狀態所以還會顯示篩選下拉鈕。 

【例 1】將使用儲存格依 A 欄篩選，準則是值大於 100，並隱藏下拉鈕： 

UsedRange.AutoFilter Field:=1, Criteria1:=">=100", VisibleDropDown:=False 

【例 2】將使用儲存格依 B 欄篩選，準則是最小值的後 3 個項目： 

UsedRange.AutoFilter Field:=2, Criteria1:= 3, Operator:=xlBottom10Items 

【例 3】將使用儲存格依 C 欄篩選，準則是最大值的前 25%項目： 

UsedRange.AutoFilter Field:=3, Criteria1:= 25, Operator:=xlTop10Percent 

【例 4】將使用儲存格依 A 欄篩選，準則是字型為紅色的項目： 

UsedRange.AutoFilter Field:=1, Criteria1:= vbRed, _ 
                                  Operator:= xlFilterFontColor 

【例 5】將使用儲存格依 B 欄篩選，準則是高於平均值的項目： 

UsedRange.AutoFilter Field:=2, Criteria1:= 33, _ 
                     Operator:=xlFilterDynamic   '33 表高於平均、34 表低於平均 

【例 6】將使用儲存格依 C 欄篩選，準則是姓氏為張、何的項目： 

UsedRange.AutoFilter Field:=3, Criteria1:= Array("張*", "何*"),  _ 
                                Operator:=xlFilterValues 

【例 7】將使用儲存格依 A 欄篩選，準則是倫敦、紐約、東京的項目： 

Dim aryC(3) As String: aryC(0)= "倫敦": aryC(1)= "紐約": aryC(2)= "東京" 
UsedRange.AutoFilter Field:=1, Criteria1:= aryC, Operator:=xlFilterValues 

【例 8】將使用儲存格依 B 欄篩選，準則是值超出 0~100 的項目： 

UsedRange.AutoFilter Field:=2, Criteria1:=" <0", Operator:=xlOr, _ 
                               Criteria2:=" >100" 

【例 9】將使用儲存格依 C 欄篩選，準則是日期為 2021 年的項目： 

UsedRange.AutoFilter Field:=3, Criteria1:=">=01/01/2021", _ 
                                 Operator:=xlAnd, Criteria2:="<=12/31/2021" 

【例 10】若工作表有使用自動篩選，就顯示全部資料： 

If  ActiveSheet.AutoFilterMode = True  Then  ActiveSheet.ShowAllData 

【例 11】若工作表有使用自動篩選，就取消自動篩選： 

If Not ActiveSheet.AutoFilter Is Nothing Then  
         ActiveSheet.AutoFilterMode = False 
End If 

 

 12.2   篩選基金績效 

：12-2 基金績效篩選.xlsm 

活頁簿中有「篩選」、「複製」、「資料」三個工作表。在「篩選」工

作表中按微調按鈕，可以設定篩選的百分比 (10 ~ 100，預設值 50)。在

「一個月」~「sharp」欄位快按兩下，會依該欄遞減排序並篩選出前指

定百分比的項目。按  鈕會將篩選後的資料複製到「複製」工作表中。

按  鈕會將「資料」工作表的原始資料，複製到「篩選」工作表。 

執行結果  

 

 上機操作  

Step 01  開啟範例「12-2 基金績效篩選(練習檔)」活頁簿檔案，內含「篩選」、

「複製」、「資料」三個工作表。 

Step 02  建立 ActiveX 控制項 

 在 N1 儲存格建立 SpinButton1 微調按鈕控制項，設定 Min = 10(最小值)、

Max = 100(最大值)、Value = 50(預設值)、SmallChange = 10(增減值)、

LinkedCell = "M1"(連結 M1 儲存格)等屬性值。 

 在 N1~O1 儲存格建立 BtnCopy、BtnUndo 兩個命令按鈕控制項。 

  

Step 03  撰寫程式碼  

【 工作表 1 程式碼 】 
01 Private Sub Worksheet_BeforeDoubleClick(ByVal Target As Range, _ 
                 Cancel As Boolean) 

02    Cancel = True                   '關閉編輯狀態 

03    Dim col As Integer 

04    col = Target.Column             '取得快按兩下的儲存格的欄數 

05    If Target.Row = 1 And (col >= 3 And col <= 11) Then 

06        Range("A:K").Sort Key1:=Cells(1, col), Order1:=xlDescending, _ 
                     Header:=xlYes 

07       Range("A:K").AutoFilter field:=col,  _ 
             Criteria1:=SpinButton1.Value, Operator:=xlTop10Percent 

08    End If 

09 End Sub 

SpinButton1 

BtnCopy 

BtnUndo 

遞減排序 

11 Private Sub BtnCopy_Click() 

12    Range("A:K").Copy Destination:=Worksheets("複製").Range("A1") 

13    Worksheets("複製").Columns("A:K").AutoFit      '自動調整大小 

14 End Sub 

15 

16 Private Sub BtnUndo_Click() 

17    If Not ActiveSheet.AutoFilter Is Nothing Then 

18        ActiveSheet.AutoFilterMode = False        '取消自動篩選 

19    End If 

20    Worksheets("資料").UsedRange.Copy _ 
                Destination:=Worksheets("篩選").Range("A1") 

21 End Sub 

說明 

1. 第 1~9行：使用者在工作表快按兩下時，會觸發 Worksheet_BeforeDoubleClick
事件。 

2. 第 3~4 行：Target 參數代表快按兩下的儲存格，用 Column 屬性取得欄數 col。 

3. 第 5~8 行：若快按兩下的儲存格在第一列且為 3~11 欄，就做排序和篩選。 

4. 第 6 行：使用 Sort 方法依 Target 儲存格做遞減排序，指定有標題列。 

5. 第 7 行：使用 AutoFilter 方法依 col 欄位做篩選，準則為前 SpinButton1.Value
百分比的項目。 

6. 第 11~14 行：按「複製」鈕時會用 Copy 方法，將篩選後的資料複製到「複

製」工作表中。然後用 AutoFit 方法自動調整 A 到 K 欄的大小。 

7. 第 16~21 行：按「復原」鈕時，若「篩選」工作表有自動篩選就取消，然後

用 Copy 方法將「資料」工作表的資料複製到「篩選」工作表。 

 12.3   快按兩下標題欄位彙整銷售資料 

：12-3 銷售資料彙整.xlsm 

在「資料」工作表的「業務員」、「月份」、「地區」、「品名」標題

欄位上快按兩下，會依該欄位的項目建立工作表，並將相關的資料複製

至對應的工作表中。 

執行結果  

 

 上機操作  

Step 01  開啟範例「12-3 銷售資料彙整(練習檔)」活頁簿檔案，內含「資料」

工作表。 

Step 02  撰寫程式碼 

【 工作表 1 程式碼 】 
01 Private Sub Worksheet_BeforeDoubleClick(ByVal Target As Range, _ 
                Cancel As Boolean) 

02    Application.Calculation = xlCalculationManual   '停止公式自動計算 

03    Application.ScreenUpdating = False  ' 停止畫面更新 

04    Application.DisplayStatusBar = False ' 停止更新狀態列 

05    Application.EnableEvents = False  ' 停止事件處理 

06    Cancel = True                            '關閉編輯狀態 

07    Dim col, fRow As Integer 

08    col = Target.Column '取得快按兩下的儲存格的欄數 

09    fRow = UsedRange.Rows.Count              '最下列數 

10    If Target.Row = 1 And (col >= 1 And col <= 4) Then 

11        Dim dic 

12        Set dic = CreateObject("scripting.dictionary")  '建立字典 

13        For Each cel In Range(Cells(2, col), Cells(fRow, col)) 

14            dic.Item(cel.Value) = 1 

15        Next 

16        Dim IsExist As Boolean         '紀錄工作表是否存在 

17        For Each d In dic.keys         '根據字典的鍵值逐一處理 

18            IsExist = False            '預設工作表不存在 

19            For i = 1 To Worksheets.Count 

產生工作表 

標題欄位快按兩下 

20                If Worksheets(i).Name = d Then 

21                    IsExist = True     '設工作表存在 

22                    Exit For 

23                End If 

24            Next 

25            If IsExist = False Then     '若工作表不存在 

26                Worksheets.Add(after:=Worksheets(Worksheets.Count)).Name = d 

27            Else 

28                Worksheets(d).Cells.Clear   '將工作表內容清空 

29            End If 

30            Range("A:G").AutoFilter field:=col, Criteria1:=d 

31            Range("A:G").Copy Destination:=Worksheets(d).Range("A1") 

32            Range("A:G").AutoFilter 

33        Next 

34    End If 

35    Application.Calculation = xlCalculationAutomatic   '恢復公式自動計算 

36    Application.ScreenUpdating = True  '恢復畫面更新 

37    Application.DisplayStatusBar = Truee  '恢復更新狀態列 

38    Application.EnableEvents = True  '恢復事件處理 

39 End Sub 

說明 

1. 第 1~39 行 ： 使 用 者 在 工 作 表 快 按 兩 下 時 ， 會 觸 發 Worksheet_ 
BeforeDoubleClick 事件。 

2. 第 2~5、35~38 行：VBA 程式執行時，Excel 也會執行其對應的功能。為避免

影響執行速度，可以先停止 Excel 的部分功能，程式執行後再恢復功能。 

3. 第 10~34 行：若快按兩下的儲存格在第一列且為 1~4 欄位才彙整資料。 

4. 第 11~12 行：建立字典 dic，字典(dictionary)是儲存「鍵」(key)和「值」(value)
成對的資料結構，一個「鍵」只會對應到一個「值」，也就是「鍵」是唯一

不能重複。因為要找出指定欄中不重複的項目，利用字典資料結構最簡便。 

5. 第 13~15 行：將指定欄中的所有儲存格值作為「鍵」，「值」為 1(可用其它

固定值)，逐一放入 dic 字典，因為「鍵」是唯一所以重複資料會自動被刪除。 

6. 第 17~34 行：逐一取出 dic 字典中所有的「鍵」值，先檢查有無以「鍵」值

為名的工作表，如果沒有就新增，如果已經存在就清空內容。然後以「鍵」

值篩選資料後，將資料複製到對應的工作表中。 

7. 第 19~24 行：逐一取得工作表的 Name 屬性值來比對「鍵」值，就能得知工

作表是否存在。 

8. 第 25~29 行：如果工作表不存在時就用 Add 方法新增，並指定 Name 屬性值

來設定名稱；存在時用 Cells 的 Clear 方法將工作表內容清空。 

9. 第 30~32 行：用 AutoFilter 方法篩選資料，然後用 Copy 方法複製資料到對應

的工作表中。最後取消自動篩選的功能。 

 

12.4   用快速鍵整理基金資料 

：12-4 基金資料整理.xlsm 

設計按快速鍵  +  +  時，會整理由網路下載的基金評等資

料。整理時先刪除空白列，新增一個「基金類別」標題欄位，然後將多

餘的標題列和基金類別列刪除。 

執行結果  

 

 

 

空白列 

空白列 

新增欄位 

移除 
空白列和 
標題列 

基金類別列 



Excel VBA 新手入門 

12-12 

 上機操作  

Step 01  開啟範例「12-4 基金資料整理(練習檔)」活頁簿檔案。 

Step 02  在 ThisWorkbook 中建立 Workbook_Open 和 Workbook_BeforeClose 
兩個事件程序。 

【 ThisWorkbook 程式碼 】 
01 Private Sub Workbook_Open() 

02    Application.OnKey "^+{b}", "ArrangeData" 

03 End Sub 

04 

05 Private Sub Workbook_BeforeClose(Cancel As Boolean) 

06    Application.OnKey "^+{b}" 

07 End Sub 

說明 

1. 第 1~3 行：想要按快速鍵就執行 Sub 程序，必須在 Workbook_Open 事件程序

中，使用 Application 物件的 OnKey 方法。OnKey 方法的引數 "^+{b}"表按  

+  +  鍵(若是 Alt 鍵則用%)，"ArrangeData"表執行的程序。注意設定

的快速鍵會蓋掉 Excel 預設的快速鍵，所以可以使用較少用的按鍵組合。 

2. 第 5~7 行：在 Workbook_BeforeClose 事件程序中，使用 OnKey 方法取消快

速鍵的設定，來恢復 Excel 預設的快速鍵。 

3. Workbook_Open 事件程序只有在開啟活頁簿時才會執行，所以寫完程式要測

試前，必須關閉活頁簿再重新開啟。 

Step 03  在活頁簿名稱上按右鍵，執行【插入/模組】指令，然後新增 Sub 程

序 ArrangeData()。 

 

1 

2 

3 

【 模組 Module1 程式碼 】 
01 Private Sub ArrangeData() 

02    Range("A:A").SpecialCells(xlCellTypeBlanks).EntireRow.Delete 

03    Range("A1").EntireColumn.Insert shift:=xlToRight    '插入 A 欄 

04    Range("A2").Value = "基金類別"               '新增標題欄位 

05    Dim fRow As Integer 

06    fRow = Range("B1").End(xlDown).Row 

07    Dim kind As String                          '紀錄基金類別 

08    For i = 1 To fRow - 2 

09        If Cells(i + 1, 2).Value = "基金名稱" Then 

10            kind = Cells(i, 2)                  '取得基金類別 

11        End If 

12        Cells(i + 2, 1).Value = kind            '寫入基金類別 

13    Next 

14    Rows(1).Delete                              '刪除第一列 

15    fRow = Range("B1").End(xlDown).Row 

16    For i = fRow To 2 Step -1 

17        If Cells(i, 2).Value = "基金名稱" Then 

18            Rows(i).Delete                      '刪除多餘的標題列 

19            Rows(i - 1).Delete                    '刪除基金類別列 

20        End If 

21    Next 

22 End Sub 

 
說明 

1. 第 1~22 行：使用者按  +  +  快速鍵，所執行的 ArrangeData 程

序要寫在模組中。 
2. 第 2 行：使用 A 欄的 SpecialCells 方法，指定引數為 xlCellTypeBlanks 取得

空白儲存格。然後用 EntireRow.Delete 方法，將空白儲存格所在列刪除，一

行程式就可以移除所有的空白列。 

3. 第 3~4 行：使用 Insert 方法插入 A 欄，然後在 A2 儲存格設定「基金類別」

標題欄位。 

4. 第 7~13 行：用 For 迴圈由第 1 列到倒數第 2 列，若下一列 B 欄儲存格值為 

"基金名稱"，就讀取 B 欄儲存格值(即基金類別)到 kind 變數，然後寫到下兩

列的 A 欄儲存格中。 

 

5. 第 16~21 行：刪除多餘的標題列時，因為列數會不斷減少，所以用 For 迴圈

由下而上刪除，程式碼會必較簡潔。 

 

12.5   用快捷功能表搜尋錯誤資料 

：12-5 搜尋錯誤資料.xlsm 

設計一個使用快捷功能表，來搜尋錯誤資料或執行指定功能。按右鍵會

出現快捷功能表，有「空白儲存格」(標記空白儲存格)、「日期錯誤」 
(標記日期錯誤儲存格)、「公式值錯誤」(標記公式值錯誤儲存格)、 
「刪除字串中空白」(刪除字串中空白字元)、「轉大寫字母」(英文字母

轉成大寫)和「轉全形字」(英、數字轉成全形字)等功能項目，執行時會

對按右鍵所在欄的資料執行指定功能。 

執行結果  

 快捷功能表 

複製 

 上機操作  

Step 01  開啟範例「12-5 搜尋錯誤資料(練習檔)」活頁簿檔案。 

Step 02  撰寫工作表 1 程式碼 

【 工作表 1 程式碼 】 
01 Private Sub Worksheet_BeforeRightClick(ByVal Target As Range, _ 
                        Cancel As Boolean) 

02    Cancel = True 

03    On Error Resume Next              '錯誤產生時跳過 

04    Application.CommandBars("myMenu").Delete    '刪除 myMenu 功能表 

05    With Application.CommandBars.Add("myMenu", msoBarPopup) 

06        With .Controls.Add    '新增功能項目 

07            .Caption = "空白儲存格"   '項目文字 

08            .OnAction = "BlankCells"    '執行程序 

09        End With 

10        With .Controls.Add 

11            .Caption = "日期錯誤" 

12            .OnAction = "ErrorDate" 

13        End With 

14        With .Controls.Add 

15            .Caption = "公式值錯誤" 

標記錯誤 
資料 

刪除字串中

空白字元 

16            .OnAction = "ErrorFormula" 

17        End With 

18        With .Controls.Add 

19            .Caption = "刪除字串中空白" 

20            .OnAction = "DelStrBlank" 

21        End With 

22        With .Controls.Add 

23            .Caption = "轉大寫字母" 

24            .OnAction = "Upper" 

25        End With 

26        With .Controls.Add 

27            .Caption = "轉全形字" 

28            .OnAction = "Wide" 

29        End With 

30        .ShowPopup                             '顯示快捷功能表 

31    End With 

32    Application.CommandBars("myMenu").Delete   '刪除 myMenu 功能表 

33    On Error GoTo 0                            '恢復錯誤機制 

34    Target.Select 

35 End Sub 

說明 

1. 第 1~35 行：使用者在工作表上按右鍵，會觸發 Worksheet_BeforeRightClick
事件，在程序中編寫顯示快捷功能表的程式碼。 

2. 第 3~4 行：使用 CommandBars 的 Delete 方法刪除 myMenu 功能表，如果該

功能表不存在時會產生錯誤，所以第 3 行設定錯誤產生時跳過錯誤敘述。 

3. 第 5~31 行：使用 CommandBars 物件的 Add 方法新增 myMenu 功能表，並在

其中新增功能項目。 

4. 第 5 行：CommandBars 的 Add 方法指定各種引數值，"myMenu"引數值表功

能表的名稱，msoBarPopup 引數值表為快捷功能表型態。 

5. 第 6~9 行：使用 Controls 的 Add 方法新增功能項目，設 Caption 屬性值指定

項目文字為「空白儲存格」，設 OnAction 屬性值指定執行 BlankCells 程序。 

6. 第 10~29 行：使用 Controls 的 Add 方法，繼續新增功能項目。 

7. 第 30 行：使用 ShowPopup 方法顯示快捷功能表。 

8. 第 32 行：使用 CommandBars 的 Delete 方法刪除 myMenu 功能表。 

9. 第 34 行：選取按右鍵的儲存格。 

Step 03  撰寫 Module1 程式碼  

【 模組 Module1 程式碼 】 
01 Private Function ColRng() As Range 

02    Dim fRow As Integer, c As Integer 

03    fRow = ActiveSheet.UsedRange.Rows.Count    '取得使用儲存格的最下列 

04    c = ActiveCell.Column                      '取得作用儲存格所在欄 

05    Set ColRng = Range(Cells(2, c), Cells(fRow, c)) 

06 End Function 

07 

08 Private Sub BlankCells() 

09    On Error Resume Next                  '錯誤產生時跳過 

10    ColRng.SpecialCells(xlCellTypeBlanks).Interior.ColorIndex = 6 

11    On Error GoTo 0                       '恢復錯誤機制 

12 End Sub 

13 

14 Private Sub ErrorDate() 

15    On Error Resume Next 

16    ColRng.SpecialCells(xlCellTypeConstants, 2).Interior.ColorIndex = 7 

17    On Error GoTo 0 

18 End Sub 

19 

20 Private Sub ErrorFormula() 

21    On Error Resume Next 

22    ColRng.SpecialCells(xlCellTypeFormulas, 16).Interior.ColorIndex = 8 

23    On Error GoTo 0 

24 End Sub 

25 

26 Private Sub DelStrBlank() 

27    ColRng.Replace What:=" ", Replacement:="", Lookat:=xlPart 

28 End Sub 

29 

30 Private Sub Upper() 

31    For Each c In ColRng 

32        c.Value = UCase(c.Value)            '使用 UCase 函數轉成大寫字母 

33    Next 

34 End Sub 

35 

36 Private Sub Wide() 

37    For Each c In ColRng 

38        c.Value = StrConv(c.Value, vbWide)  '使用 StrConv 函數轉成全形字 

39    Next 

40 End Sub 

說明 

1. 第 1~40 行：功能項目指定執行的程序，要寫在模組(Module)中。 
2. 第 1~6 行：ColRng 自定 Function 程序會傳回作用儲存格所在欄有資料的儲存

格範圍。 

3. 第 8~12 行：BlankCells 程序使用 SpecialCells(xlCellTypeBlanks)方法選取空

白儲存格，然後設定儲存格的背景色。因為若沒有空白儲存格執行時會產生

錯誤，所以要先設定錯誤產生時跳過。 

4. 第 14~18 行：ErrorDate 程序使用 SpecialCells(xlCellTypeConstants, 2)方法選

取日期錯誤的儲存格，然後設定儲存格的背景色。 

5. 第 20~24 行：ErrorFormula 程序使用 SpecialCells(xlCellTypeFormulas, 16)
方法選取錯誤公式值的儲存格，然後設定儲存格的背景色。 

6. 第 26~28 行：DelStrBlank 程序使用 Replace 方法，逐一取代儲存格中的空白

字元為空字串，來達成刪除空白字元的效果。 

7. 第 30~34 行：Upper 程序使用 Ucase 函數，在 For Each 迴圈中逐一將儲存格

中的字母轉為大寫。 

8. 第 36~40 行：Wide 程序使用 StrConv 函數引數值設為 vbWide，在 For Each 迴

圈中逐一將儲存格中的英數字轉為全形字。 

